
Computer Tutorial 2

The purpose of this tutorial is to introduce subspaces of vector spaces, to demon-
strate how MAGMA can be used to solve matrix equations and how to use matrices
to calculate projections and lines of best fit.
The MAGMA commands you will learn in this tutorial are:

sub KMatrixSpace Dimension Transpose Solution

1. In MAGMA, create the subspace of (R5)′ spanned by the vectors (1,1,0,1,1),
(0,1,0,1,−1) and (3,1,0,1,5). Do this by carrying out the following steps:

(i) Define the real field R := RealField();

(ii) Create a vector space V of dimension 5 over R (as in Computer Tuto-
rial 1).

(iii) Declare the vectors above as a1, a2 and a3. For example,
a1 := V![1,1,0,1,1];

(iv) Create the subspace W spanned by a1, a2 and a3 using the command
W := sub< V | a1, a2, a3 >;

(v) Find the dimension of W. (Magma has a function Dimension.)
(vi) Are the vectors a1, a2 and a3 linearly independent?

Solution.

> R := RealField();
> V := VectorSpace(R,5);
> a1 := V![1,1,0,1,1];
> a2 := V![0,1,0,1,-1];

> a3 := V![3,1,0,1,5];
> W := sub< V | a1, a2, a3 >;
> Dimension(W);

Since they span a two-dimensional space rather than a three-dimensional
space, a1, a2 and a3 linearly dependent.

2. Consider the following vectors in R5:

b
˜
1 =


1
1
0
1
1

 , b
˜
2 =


0
1
0
1
0

 , b
˜
3 =


−1
2
−3
4
−5

 , v
˜

=


1
5
−4
3
6

 .

We wish to calculate the projection of v
˜

onto the subspace W of R5 with
basis {b

˜
1, b

˜
2, b

˜
3}. Since MAGMA uses row vectors rather than column vectors,

we have to transpose everything.

(i) We have already (in Exercise 1) created the field R and V, a 5-dimensional
space over R. Define b1 :=V![1,1,0,1,1]; and similarly define b2, b3
and v to be the transposes of b

˜
2, b

˜
3 and v

˜
above.

(ii) Create the vector space of 3× 5 matrices using the command
M := KMatrixSpace(R,3,5);

(iii) Create the matrix B with rows b1, b2 and b3 using the command
B := M![b1,b2,b3];

(This is the transpose of the matrix A whose columns are b
˜
1, b

˜
2 and b

˜
3.)

(iv) Print B and v to check that you have entered everything correctly, and
then define A := Transpose(B);.

(v) According to the theory developed in lectures, the projection of v
˜

onto W
is the vector Ax

˜
, where AT Ax

˜
= AT v

˜
. Taking transposes this equation

becomes x
˜

T AT A = v
˜

T A, since transposing reverses multiplication. We
can obtain the vector x

˜
T via the command

x := Solution(B*A,v*A);

(Explanation: if M is a matrix and b a (row) vector then Solution(M,b)
is a vector x that is a solution of the matrix equation xM = b.)

(vi) The projection is p
˜

= Ax
˜
. Transposing (remembering that B = AT and

that the vector x that MAGMA has found is actually x
˜

T ), we see that
x * B;

will print out the row vector that is the transpose of p
˜
.

Solution.

> b1 := V![1,1,0,1,1];
> b2 := V![0,1,0,1,0];
> b3 := V![-1,2,-3,4,-5];
> M := KMatrixSpace(R,3,5);
> B := M![b1,b2,b3];
> v := V![1,5,-4,3,6];
> B,v;
[1 1 0 1 1]
[0 1 0 1 0]
[-1 2 -3 4 -5]

(1 5 -4 3 6)
> A := Transpose(B);
> x := Solution( B*A,v*A );
> x;
(7/2 1/2 0)
> x * B;
(7/2 4 0 4 7/2)

3. Repeat the previous exercise to find the projection of (1, 0, 1, 0)T onto the
subspace of R4 spanned by (−2, 1, 0, 1)T and (−3, 0, 1, 0)T .

Solution.

> V := VectorSpace(R,4);
> b1 := V![-2,1,0,1];
> b2 := V![-3,0,1,0];
> v := V![1,0,1,0];
> M := KMatrixSpace(R,2,4);

> B := M![b1,b2];
> x := Solution( B*A,v*A )
where A is Transpose(B);

> x * B;
(2/3 -1/3 0 -1/3)

4. Find the least squares line of best fit for the following data.

atomic no. log10 of half-life (seconds)
104 1.8
105 1.5
106 −0.05
107 −0.97
108 −2.7
109 −2.5

There are six data points (x1, y1), . . . , (x6, y6). By the theory given in lectures
the line of best fit is y = a+bx, where a(1, 1, 1, 1, 1, 1)+b(x1, x2, x3, x4, x5, x6)
is as close as possible to (y1, y2, y3, y4, y5, y6). Use the following commands:

R := RealField(10);
V := VectorSpace(R,6);
B := M![1,1,1,1,1,1,104,105,106,107,108,109] where

M is KMatrixSpace(R,2,6);
y := V![1.8,1.5,-0.05,-0.97,-2.7,-2.5];
Solution(B*A,y*A) where A is Transpose(B);

Now make a prediction for the half-life of element number 110.

Solution.

> V := VectorSpace(R,6);
> M := KMatrixSpace(R,2,6);
> B := M![1,1,1,1,1,1,104,

105,106,107,108,109];
> y := V![1.8,1.5,-0.05,

-2.7,-2.5];
> x := Solution( B*A,y*A )

where A is Transpose(B);
> x;
(106.07403517 -1.00056997)
> U := VectorSpace(R,2);

> a := U![1,110];
> InnerProduct(x,a);
-3.98866153
> 10∧(-3.98866153);
0.000102645 . . .

Recall that the value of x gives the coefficients α and β in the equation
y = α + βx of the line of best fit. So in this case the line is y = 106.074− x.
Evaluating the right hand-side at x = 110 amounts to calculating the dot
product of (106.074,−1) and (1, 110), and this explains the three MAGMA

commands above following the calculation of x. Note that −3.988 . . . is the
log of the half-life; to get the half-life itself we must raise 10 to this power.

If the prediction is correct, the half-line is about one ten-thousandth of a
second.

5. The data below gives the concentration of atmospheric CO2 (in ppm) in var-
ious years. Find the parabola of best fit.

Year Concentration
1852 288
1914 300
1932 306
1950 311
1960 317.9
1970 325.6
1980 338.7
1990 354.2

Make a prediction for the year 2040.
If a change of 100 ppm corresponds to a change of 1◦C in the average global
temperature, how much warmer will it be then? (Note: In the last ice age the
mean global temperature was about 1◦C less than it is now.)
(The last two questions are derived from articles that appeared in “Scientific
American”, April 1989.)

Solution.

> V := VectorSpace(R,8);
> M := KMatrixSpace(R,3,8);
> B := M![1,1,1,1,1,1,1,1,1852,1914,1932,1950,1960,1970,1980,
1990,1852∧2,1914∧2, 1932∧2,1950∧2,1960∧2,1970∧2,1980∧2,1990∧2];

> y := V![288,300,306,311,317.9,325.6,338.7,354.2];
> x := Solution( B*A,y*A ) where A is Transpose(B);
> x;
(16070.0948 -16.85201478 0.004498490307)
> P<t>:=PolynomialAlgebra(R);
> f:=P![x[i]: i in [1,2,3]];
> f;
0.004498490307*t∧2 - 16.85201478*t + 16070.0948
> Evaluate(f,2040);
412.9019

A change of 59 ppm as compared with 2000 means—if this theory is to be
believed—that in 2040 things will be about about a half a degree warmer than
in 2000.


